CHECKING THE PAPER
Weighted Partition Identities and Divisor Sums
05.16.16 and updated 03.23.17

```
> with(qseries):
> currentdir("H:\\math\\research\\new-fokkink");
```

"H:\math\research\new-fokkink" (1)

```
> ##You will need to change this to directory where you saved
> ##the program WPROGS.txt
```

```
> read "WPROGS.txt":
```

Warning, `x2` is implicitly declared local to procedure `ffwstatz1`

```
> fftable(10);
```

```
[1, 2, 3, 4], -1
[2, 3, 5], 2
[1, 4, 5], 1
[1, 3, 6], 1
[4, 6], -4
[1, 2, 7], 1
[3, 7], -3
[2, 8], -2
[1, 9], -1
[10], 10
```

```
> seq(ffwstat(n),n=1..10);
```

$1, 2, 2, 3, 2, 4, 2, 4, 3, 4$ (2)

```
> seq(ffwstat2(n),n=1..10);
```

$1, -1, 2, 0, 1, 0, 3, 2, 2, 2$ (3)

```
> Y:=1+add(ffwstat2(n)*q^n,n=1..10);
```

$$Y := -10\,q^{10} + 7\,q^9 - 4\,q^8 + 10\,q^7 - 6\,q^6 + 4\,q^5 - 3\,q^4 + 4\,q^3 - 2\,q^2 + q + 1 \quad (4)$$

```
> prodmake((Y-1)/q,q,9,list);
```

$[2, -3, -3, -1, 4, 6, -8, -18]$ (5)

```
> seq(ffwstat2(n)*(-1)^(n-1),n=1..10);
```

$1, 2, 4, 3, 4, 6, 10, 4, 7, 10$ (6)

THEOREM 1.1 (Fokkink, Fokkink and Wang)

```
> seq(ffwstat(n),n=1..10);
```

$1, 2, 2, 3, 2, 4, 2, 4, 3, 4$ (7)

```
> seq(numdivs(n),n=1..10);
```

$1, 2, 2, 3, 2, 4, 2, 4, 3, 4$ (8)

```
> {seq(ffwstat(n)-numdivs(n),n=1..40)};
```

$\{0\}$ (9)

ANALYTIC FORM

780 (10)

```
> LFFW:=add( (-1)^(n-1)*q^(n*(n+1)/2)/aqprod(q,q,n)/(1-q^n),n=1.
  .40):
> RFFW:=add(q^n/(1-q^n),n=1..700):
```

```
> series(LFFW-RFFW,q,700);
```

$$O(q^{700}) \tag{11}$$

THEOREM 1.2

```
> LTH12a:=add( (-1)^(n-1)*q^(n^2)*z^n/aqprod(z*q,q^2,n)/(1-z*q^(2*
  n)),n=1..10):
> RTH12a:=add( z^n*q^(n*(n+1)/2)*aqprod(q,q,n-1)/aqprod(z*q,q,n),n=
  1..20):
> normal(series(LTH12a-RTH12a,q,100));
```

$$O(q^{100}) \tag{12}$$

```
> LTH12b:=add( (-1)^(n)*q^(n^2+n)*z^n/aqprod(z*q^2,q^2,n)/(1-z*q^
  (2*n+1)),n=0..10):
> RTH12b:=add( z^n*q^(n*(n+1)/2)*aqprod(q,q,n)/aqprod(z*q,q,n),n=0.
  .20):
> normal(series(LTH12b-RTH12b,q,100));
```

$$O(q^{100}) \tag{13}$$

COROLLARY 1.3

(i)

```
> omega1a := ptn -> (-1)^( (lop(ptn) -1)/2):
> sumomega1a := n->add( omega1a(ptn), ptn in select(PC1,partition
  (n))):
> selodd:=proc(n) if modp(n,2)=1 then RETURN(true): else RETURN
  (false): fi: end:
> d1:=n->nops(select(selodd,divisors(n))):

> seq(sumomega1a(n),n=1..10);
```

$$1, 1, 2, 1, 2, 2, 2, 1, 3, 2 \tag{14}$$

```
> seq(d1(n),n=1..10);
```

$$1, 1, 2, 1, 2, 2, 2, 1, 3, 2 \tag{15}$$

```
> {seq(sumomega1a(n)-d1(n),n=1..40)};
```

$$\{0\} \tag{16}$$

(ii)

```
> omega1b := ptn -> (-1)^( (lop(ptn) -1)/2 + nops(ptn)+1):
> sumomega1b := n->add( omega1b(ptn), ptn in select(PC1,partition
  (n))):
> d18:=proc(n)
> local divs,d,c:
>   divs:=divisors(n):
>   c:=0:
>   for d in divs do
>     if member(modp(d,8),{1,7}) then c:=c+1: fi:
>     if member(modp(d,8),{3,5}) then c:=c-1: fi:
>   od:
> RETURN(c):
> end:

> b:=n->(-1)^(n*(n-1)/2)*d18(n);
```

$$b := n \rightarrow (-1)^{\frac{1}{2} n (n-1)} d18(n) \tag{17}$$

```
> seq(sumomega1b(n),n=1..10);
```

(18)

$$1, -1, 0, 1, 0, 0, -2, 1, 1, 0 \quad (18)$$

```
> seq(b(n),n=1..10);
```

$$1, -1, 0, 1, 0, 0, -2, 1, 1, 0 \quad (19)$$

```
> {seq(sumomega1b(n)-b(n),n=1..40)};
```

$$\{0\} \quad (20)$$

Analytic Form of (ii)

```
> Lii0:=add( q^(n^2)/aqprod(-q,q^2,n)/(1+q^(2*n)),n=1..20):
> Lii:=add( (-1)^(n+1)*q^(n*(n+1)/2)*aqprod(q,q,n-1)/aqprod(-q,q,
  n),n=1..20):
> 20*19/2;
```

$$190 \quad (21)$$

```
> Rii:=add(b(n)*q^n,n=1..400):
> series(Lii-Rii,q,190);
```

$$O(q^{231}) \quad (22)$$

```
> series(Lii0-Rii,q,400);
```

$$O(q^{400}) \quad (23)$$

(iii)

```
> omega2a := ptn -> (-1)^( lep(ptn)/2 ):
> sumomega2a := n->add( omega2a(ptn), ptn in select(PC2,partition
  (n))):
> seq([n,sumomega2a(n)],n=1..10);
```

$$[1, 1], [2, 0], [3, 1], [4, 0], [5, 0], [6, 1], [7, 0], [8, 0], [9, 0], [10, 1] \quad (24)$$

```
> add(sumomega2a(n)*q^n,n=1..40);
```

$$q^{36} + q^{28} + q^{21} + q^{15} + q^{10} + q^6 + q^3 + q \quad (25)$$

Analytic Form of (iii)

```
> Liii:=add( (-1)^(n)*q^(n^2+n)/aqprod(q^2,q^2,n)/(1-q^(2*n+1)),n=
  0..20):
> series(Liii,q,10);
```

$$1 + q + q^3 + q^6 + O(q^{10}) \quad (26)$$

```
> Riii:=add(q^(n*(n+1)/2),n=0..40):
> series(Liii-Riii,q,400);
```

$$O(q^{400}) \quad (27)$$

(iv)

```
> omega2b := ptn -> (-1)^( lep(ptn)/2 + nops(ptn) ):
> sumomega2b := n->add( omega2b(ptn), ptn in select(PC2,partition
  (n))):
> seq(sumomega2b(n),n=1..10);
```

$$-1, 2, -1, 0, -2, 3, 0, 0, -2, 1 \quad (28)$$

```
> seq((-1)^n*d18(8*n+1),n=1..10);
```

$$-1, 2, -1, 0, -2, 3, 0, 0, -2, 1 \quad (29)$$

```
> {seq(sumomega2b(n)-(-1)^n*d18(8*n+1),n=1..40)};
```

$$\{0\} \quad (30)$$

Analytic Form of (iv)

```
> Liv:=add( q^(n^2+n)/aqprod(-q^2,q^2,n)/(1+q^(2*n+1)),n=0..20):
> series(Liv,q,10);
```

$$1 - q + 2q^2 - q^3 - 2q^5 + 3q^6 - 2q^9 + O(q^{10}) \quad (31)$$

```
> Riv:=add((-1)^n*d18(8*n+1)*q^n,n=0..400):
> series(Liv-Riv,q,400);
```

$$O(q^{400}) \quad (32)$$

==========

Example 1.4

```
> read "WPROGS.txt":
> PC19:=select(PC1,partition(9)):
> map(ptn2freqs,PC19);
```

$$[[1^9], [1^7, 2], [1^5, 2^2], [1^3, 2^3], [1, 2^4], [1^6, 3], [1^3, 3^2], [1^2, 3, 4], [1, 3, 5]] \quad (33)$$

```
> for ptn in PC19 do
> print(ptn2freqs(ptn),lop(ptn),nops(ptn),omega1a(ptn),omega1b(ptn)
  );od;
```

$$[1^9], 1, 9, 1, 1$$
$$[1^7, 2], 1, 8, 1, -1$$
$$[1^5, 2^2], 1, 7, 1, 1$$
$$[1^3, 2^3], 1, 6, 1, -1$$
$$[1, 2^4], 1, 5, 1, 1$$
$$[1^6, 3], 3, 7, -1, -1$$
$$[1^3, 3^2], 3, 5, -1, -1$$
$$[1^2, 3, 4], 3, 4, -1, 1$$
$$[1, 3, 5], 5, 3, 1, 1 \quad (34)$$

Example 1.5

```
> PC215:=select(PC2,partition(15)):
> map(ptn2freqs,PC215);
```

$$[[1^{15}], [2^6, 3], [2^3, 3^3], [2^3, 4, 5], [2, 4^2, 5]] \quad (35)$$

```
> for ptn in PC215 do
> print(ptn2freqs(ptn),lep(ptn),nops(ptn),omega2a(ptn),omega2b(ptn)
  );od;
```

$$[1^{15}], 0, 15, 1, -1$$
$$[2^6, 3], 2, 7, -1, 1$$
$$[2^3, 3^3], 2, 6, -1, -1$$
$$[2^3, 4, 5], 4, 5, 1, -1$$
$$[2, 4^2, 5], 4, 4, 1, 1 \quad (36)$$

```
> d18(11^2);
```

$$1 \quad (37)$$

==========

PROOF OF THEOREM 1.2

Proof of (1.3)

```
> L1:=add(z^n*q^(n*(n+1)/2)*aqprod(q,q,n-1)/aqprod(z*q,q,n),n=1.
  .20):
> R1:=z*q*add(z^n*q^n*aqprod(q^2,q^2,n)/aqprod(z*q^2,q^2,n+1),n=0.
```

```
   .100):
> normal(series(L1-R1,q,100));
```

$$O(q^{100}) \tag{38}$$

```
> L2:=1 + (1-z)*add( (-1)^n*z^n*q^(n^2)/aqprod(z*q,q^2,n)/(1-z*q^
  (2*n)),n=1..20):
> R2:=(1-z)*add(aqprod(q,q^2,n)*z^n/aqprod(z*q,q^2,n),n=0..50):
> normal(series(L2-R2,z,20));
```

$$O(z^{20}) \tag{39}$$

```
> GASP39:=(a,b,c,d,e)->P1(e/a,d*e/b/c)/P2(e,d*e/a/b/c)*PHI32TOP(a,
  d/b,d/c)/PHI32BOT(d,d*e/b/c)*POW(e/a);
```

$$GASP39 := (a, b, c, d, e) \to \frac{P1\left(\frac{e}{a}, \frac{d\,e}{b\,c}\right) PHI32TOP\left(a, \frac{d}{b}, \frac{d}{c}\right) POW\left(\frac{e}{a}\right)}{P2\left(e, \frac{d\,e}{a\,b\,c}\right) PHI32BOT\left(d, \frac{d\,e}{b\,c}\right)} \tag{40}$$

```
> GASP39(q^2,1/a,z,z*q,z*q^2);
```

$$\frac{P1(z, z\,q^3\,a)\ PHI32TOP(q^2, z\,q\,a, q)\ POW(z)}{P2(z\,q^2, z\,q\,a)\ PHI32BOT(z\,q, z\,q^3\,a)} \tag{41}$$

```
> L3:=add( (-1)^n*z^n*q^(n^2)/aqprod(z*q,q^2,n)/(1-z*q^(2*n)),n=1.
  .20):
> R3:=(-1)/(1-z)+add(aqprod(q,q^2,n)*z^n/aqprod(z*q,q^2,n),n=0..50)
  :
> normal(series(L3-R3,z,20));
```

$$O(z^{20}) \tag{42}$$

```
> L4:=add( z^n*q^(n)*aqprod(q^2,q^2,n-1)/aqprod(z*q^2,q^2,n),n=1.
  .20):
> R4:=(z)/(1-z)-add(aqprod(q,q^2,n)*z^n/aqprod(z*q,q^2,n),n=1..50):
> normal(series(L4-R4,z,20));
```

$$O(z^{20}) \tag{43}$$

```
> R5a:=add(add(z^n*q^n*aqprod(q^2,q^2,n-1)*qbin(q^2,j,n+j-1)*z^j*q^
  (2*j),j=0..20),n=1..20):
> normal(series(L4-R5a,z,20));
```

$$O(z^{20}) \tag{44}$$

```
> R5b:=add(add(z^m*q^(2*m-n)*aqprod(q^2,q^2,m-1)/aqprod(q^2,q^2,m-
  n),n=1..m),m=1..20):
> normal(series(L4-R5b,z,20));
```

$$O(z^{20}) \tag{45}$$

```
> R5c:=add(add(z^m*q^(m+k)*aqprod(q^2,q^2,m-1)/aqprod(q^2,q^2,k),k=
  0..m-1),m=1..20):
> normal(series(L4-R5c,z,20));
```

$$O(z^{20}) \tag{46}$$

```
> L6:=add(aqprod(q,q^2,n)*z^n/aqprod(z*q,q^2,n),n=1..50):
> R6a:=add(add(z^(n+j)*q^(j)*qbin(q^2,j,n+j-1)*aqprod(q,q^2,n),j=0.
  .20),n=1..20):
> normal(series(L6-R6a,z,20));
```

$$O(z^{20}) \tag{47}$$

```
> R6b:=add(add(aqprod(q,q^2,n)*aqprod(q^2,q^2,m-1)*z^m*q^(m-n)
```

```
  /aqprod(q^2,q^2,m-n)/aqprod(q^2,q^2,n-1),n=1..m),m=1..20):
> normal(series(L6-R6b,z,20));
```

$$O(z^{20}) \tag{48}$$

```
> L7:=m->add(q^(m+k)/aqprod(q^2,q^2,k),k=0..m-1):
> R7:=m->1/aqprod(q^2,q^2,m-1) - add(aqprod(q,q^2,n)*q^(m-n)/aqprod
  (q^2,q^2,m-n)/aqprod(q^2,q^2,n-1),n=1..m):
> seq(normal(L7(m)-R7(m)),m=1..30);
```

$$0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0 \tag{49}$$

```
> L8:=add(add(z^m*q^(m+k)/aqprod(q^2,q^2,k),k=0..m-1),m=1..40):
> R8:=z*q/(1-z*q)/aqprod(z*q^2,q^2,25):
> normal(series(L8-R8,q,40));
```

$$O(q^{40}) \tag{50}$$

```
> L9:=add(add(z^m*q^(m-n)*aqprod(q,q^2,n)/aqprod(q^2,q^2,m-n)
  /aqprod(q^2,q^2,n-1),n=1..m),m=1..20):
> R9:=z*(1-q)/(1-z*q)/aqprod(z,q^2,30):
> R9b:=add(z^n/aqprod(q^2,q^2,n),n=0..20)*z*(1-q)/(1-z*q):
> normal(series(L9-R9b,z,20));
```

$$O(z^{20}) \tag{51}$$

```
> series(normal(series(R9-R9b,z,20)),q,20);
```

$$O(z^{20}) + O(q^{60}) \tag{52}$$

```
> L10:=z*q*(1-z)/(1-z*q):
> R10a:=z*( (1-z*q) - (1-q))/(1-z*q):
> R10b:=z - z*(1-q)/(1-z*q):
> normal(L10-R10a);
```

$$0 \tag{53}$$

```
> normal(L10-R10b);
```

$$0 \tag{54}$$

Proof of (1.4)

```
> L11:=add( (-1)^n*z^n*q^(n^2)/aqprod(z*q,q^2,n)/(1-z*q^(2*n)),n=0.
  .10):
> R11:=add( z^n*q^(n*(n-1)/2)*aqprod(q,q,n)/aqprod(z,q,n),n=0..20):
> normal(series(L11-R11,q,50));
```

$$O(q^{50}) \tag{55}$$

```
> L12:= (1-z*q) - z*q*(1-q):
> R12:=(1-z*q)*(1-q) + q*(1-z):
> normal(L12-R12);
```

$$0 \tag{56}$$

==========

Proof of Corollary 1.3

```
> omega1a := ptn -> (-1)^( (lop(ptn) -1)/2):
> sumomega1a := n->add( omega1a(ptn), ptn in select(PC1,partition
  (n))):
> L31:=add(sumomega1a(n)*q^n,n=1..40);
```

$$L31 := 2q^{40} + 4q^{39} + 2q^{38} + 2q^{37} + 3q^{36} + 4q^{35} + 2q^{34} + 4q^{33} + q^{32} + 2q^{31} + 4q^{30} \tag{57}$$
$$+ 2q^{29} + 2q^{28} + 4q^{27} + 2q^{26} + 3q^{25} + 2q^{24} + 2q^{23} + 2q^{22} + 4q^{21} + 2q^{20} + 2q^{19}$$
$$+ 3q^{18} + 2q^{17} + q^{16} + 4q^{15} + 2q^{14} + 2q^{13} + 2q^{12} + 2q^{11} + 2q^{10} + 3q^{9} + q^{8} + 2q^{7}$$

$$+2q^6+2q^5+q^4+2q^3+q^2+q$$

```
> R31a:=add(q^(n*(n+1)/2)/(1-q^n),n=1..10):
> series(L31-R31a,q,40);
```

$$O(q^{40}) \tag{58}$$

```
> R31b:=add(q^n/(1-q^(2*n)),n=1..40):
> series(L31-R31b,q,40);
```

$$O(q^{40}) \tag{59}$$

```
**********
> omega1b := ptn -> (-1)^( (lop(ptn) -1)/2 + nops(ptn)+1):
> sumomega1b := n->add( omega1b(ptn), ptn in select(PC1,partition
  (n))):
> L32:=add(sumomega1b(n)*q^n,n=1..40);
```

$$L32 := q^{36}-2q^{34}+q^{32}-2q^{31}+2q^{28}+q^{25}-2q^{23}-q^{18}+2q^{17}+q^{16}-2q^{14}+q^9+q^8 \tag{60}$$
$$-2q^7+q^4-q^2+q$$

```
> with(qseries):
> R32:=add((-1)^(n-1)*q^(n*(n+1)/2)*aqprod(q,q,n-1)/aqprod(-q,q,n),
  n=1..20):
> series(L32-R32,q,41);
```

$$O(q^{41}) \tag{61}$$

```
> Lo1:=add(add((-1)^(n+j)*q^(2*n^2-j^2),j=-n+1..n),n=1..10):
> series(R32-Lo1,q,100);
```

$$O(q^{100}) \tag{62}$$

```
> Lothm11:=add((-1)^(n*(n-1)/2)*d18(n)*q^n,n=1..100):
> series(R32-Lothm11,q,100);
```

$$O(q^{100}) \tag{63}$$

```
******
> omega2a := ptn -> (-1)^( lep(ptn)/2 ):
> sumomega2a := n->add( omega2a(ptn), ptn in select(PC2,partition
  (n))):
> combzid2a:=add( (-1)^n*z^n*q^(n^2+n)/aqprod(z*q^2,q^2,n)/(1-z*q^
  (2*n+1)),n=0..10):
> Zsumomega2a := n->add( z^(nops(ptn))*omega2a(ptn), ptn in select
  (PC2,partition(n))):
> Zsum2a:=1+add(Zsumomega2a(n)*q^n,n=1..40):
> normal(series(Zsum2a-combzid2a,q,40));
```

$$O(q^{40}) \tag{64}$$

```
*******
> omega2b := ptn -> (-1)^( lep(ptn)/2 + nops(ptn) ):
> sumomega2b := n->add( omega2b(ptn), ptn in select(PC2,partition
  (n))):
> seq(sumomega2b(n),n=1..10);
```

$$-1, 2, -1, 0, -2, 3, 0, 0, -2, 1 \tag{65}$$

```
> seq((-1)^n*d18(8*n+1),n=1..10);
```

$$-1, 2, -1, 0, -2, 3, 0, 0, -2, 1 \tag{66}$$

```
> Lcoriv:=1+add(sumomega2b(n)*q^n,n=1..40):
> Rcoriv:=add( (-1)^n*q^(n*(n+1)/2)*aqprod(q,q,n)/aqprod(-q,q,n),n=
```

```
  0..20):
> series(Lcoriv-Rcoriv,q,40);
```

$$O(q^{40}) \tag{67}$$

```
> RcorivB:=add((-1)^n*d18(8*n+1)*q^n,n=0..100):
> series(Rcoriv-RcorivB,q,100);
```

$$O(q^{100}) \tag{68}$$